Inserção dos Conceitos de Interfaces e Classes Abstratas no Framework Utilizado em Paradigmas de Linguagens de Programação
Carlos Eduardo Pontual, Fernanda d’Amorim
Filipe Motta, Leopoldo Teixeira
Universidade Federal de Pernambuco

Centro de Informática

{ceplc, frsa, famb, lmt}@cin.ufpe.br

1. Introdução

O curso Paradigmas de Linguagens de Programação tem como processo de ensino a construção de uma família de linguagens (framework) baseada nos principais paradigmas existentes. Esta metodologia proporciona aos alunos a fixação de conceitos através do entendimento do projeto de cada linguagem, bem como através do processo de extensão das linguagens já presentes no framework, implementando conceitos ainda não contemplados. 

Após uma análise da linguagem orientada a objetos (O.O.1) e dos trabalhos de turmas anteriores que a estendem, foi verificado que os conceitos de Interface e Classe Abstrata não são abordados em nenhum momento.

Segundo a definição, uma interface é um contrato entre uma classe e o mundo ao seu redor. Quando uma classe implementa uma interface, ela se compromete em prover o comportamento publicado pela interface, ou seja, ela deve implementar todos os métodos definidos por esta interface. Classes abstratas são classes que não possuem instâncias. Normalmente, esses tipos de classes possuem métodos abstratos: métodos que são declarados, mas que não são implementados, ou seja, são assinaturas de métodos. Uma classe abstrata requer que suas subclasses concretas forneçam implementações para os métodos abstratos que são definidos nela.

Como muitas linguagens permitem a implementação de interfaces (apesar de apenas algumas possuirem construções específicas para esse fim) e classes abstratas, optamos por focar nosso trabalho na inclusão desses dois conceitos no framework utilizado na disciplina de Paradigmas de Linguagens de Programação. 

No restante desse documento iremos detalhar os objetivos desse trabalho (seção 2) e exibiremos uma proposta inicial da BNF que será utilizada nesse projeto (seção 3).

2. Objetivo

Inicialmente, foi cogitada a hipótese de estender a linguagem O.O.1, inserindo os conceitos de interface e classe abstrata. Entretanto, após uma análise mais detalhada dos trabalhos realizados por turmas anteriores, foi verificado que a linguagem O.O.2, desenvolvida por alunos do período 2006.1, propõe o conceito de herança, que será necessário no caso da existência de classes abstratas. 

Assim, este projeto tem como objetivo inserir os conceitos de interface e classe abstrata na linguagem O.O.2.

O escopo do projeto inclui a redefinição da BNF proposta na linguagem O.O.2, para que ela seja capaz de prover os conceitos de interface e classe abstrata, e a implementação de tais conceitos baseada nos recursos da linguagem Java 1.5. Como não foi possível fazer uma análise mais detalhada da dificuldade do projeto proposto, o escopo inicial possui algumas limitações, como a restrição de uma classe poder implementar apenas uma interface e a obrigação de uma classe abstrata possuir pelo menos um método abstrato. 

Ao final do curso, a linguagem O.O.2 terá suporte a:

Classes abstratas com pelo menos um método abstrato;

Interfaces;

Implementação de uma classe abstrata por uma classe concreta;

Herança simples entre interfaces;

Herança entre classes abstratas;

Implementação de uma ou mais interfaces por uma classe concreta;

Métodos abstratos.

Em uma tentativa de medir as modificações e o esforço inicial necessários para a conclusão do projeto, foi elaborada uma BNF baseada na utilizada em O.O.2. Tal BNF é apresentada na próxima sessão.

3. Proposta Inicial – BNF da linguagem O.O.2 Estendida

A seguir é apresentada a BNF da proposta inicial de extensão da linguagem O.O.2. Note que as partes destacadas em negrito são correspondentes as alterações efetuadas na BNF referente à O.O.2.

Programa ::= "{" Declaracoes ";" Comando "}"


Declaracoes ::= DecInterface 

| DecClasse 

| Declaracoes “;” Declaracoes

Comando ::= AtribuicaoOO2
                      | ComDeclaracao
                      | While
                      | IfThenElse
                      | IO
                      | Comando ";" Comando
                      | Skip
                      | New
                      | ChamadaMetodoOO2
 

Skip ::=

 ComDeclaracao :: = "{" DecVariavel ";" Comando"}"

 While ::= "while" Expressao do" "{" Comando"}"

 IfThenElse ::= "if" Expressao "then" "{" Comando"}"
                        | "if" Expressao "then" "{" Comando"}" "else" "{" Comando"}"

 ChamadaMetodoOO2 ::= Expressao "." Id "(" ListaExpressao ")" 

| Expressao "." Id "("")"

 ListaExpressao ::= Expressao 

| Expressao "," ListaExpressao

 New ::= LeftExpression ":=" "new" Id

 AtribuicaoOO2 ::= LeftExpression ":=" Expressao

 IO ::= "write" "(" Expressao ")" 

| "read" "(" Id" )"

 Expressao ::= Valor 

| ExpUnaria 

| ExpBinaria 

| LeftExpression 

| this 

| super 

| Cast

 Cast ::= LeftExpression ":=" "(" TipoClasse ")" Expressao

 Valor ::= ValorConcreto

 ValorConcreto ::= ValorInteiro 

| ValorBooleano 

| ValorString 

| ValorNull



ExpUnaria ::= "-" Expressao 

| "not" Expressao 

| "length" Expressao


 ExpBinaria ::= Expressao "+" Expressao

| Expressao "-" Expressao
| Expressao "and" Expressao

| Expressao "or" Expressao

| Expressao "==" Expressao
| Expressao "++" Expressao
| Expressao "instanceof" Expressao

 

LeftExpression ::= Id AcessoAtributoOO2

 AcessoAtributoOO2 ::= LeftExpression.Id 

| "this." Id 

| "super." Id
   
 DecInterface ::= "interface" Id ["extends" Id] "{" ListaAssinaturasMetodos "}"

 ListaAssinaturasMetodos ::= AssinaturaMetodo 

| AssinaturaMetodo "," ListaAssinaturasMetodos 

AssinaturaMetodo ::= modAcesso "proc" Id "(" [ListaDeclaracaoParametro] ")"

DecClasse ::= DecClasseAbstrata 

| DecClasseConcreta

DecClasseAbstrata ::= "abstract" "classe" Id ["extends" Id] "{" CorpoClasseAbstrata "}"

CorpoClasseAbstrata ::= DecVariavel 

| DecProcedimento 

| CorpoClasseAbstrata";" CorpoClasseAbstrata 

DecClasseConcreta ::= "classe" Id ["extends" Id] ["implements" Id] "{" CorpoClasseConcreta "}"
    
CorpoClasseConcreta ::= DecVariavel 

| ListaDecProcConcreto 

| CorpoClasseConcreta ";" CorpoClasseConcreta



DecVariavel ::= Tipo Id "=" Expressao

| DecVariavel "," DecVariavel

| Tipo Id ":=" "new" Id


DecProcedimento ::= DecProcConcreto 

| DecProcAbstrato 

| DecProcedimento "," DecProcedimento

DecProcConcreto ::= AssinaturaMetodo "{" Comando "}"


ListaDecProcConcreto ::= DecProcConcreto 

| DecProcConcreto "," ListaDecProcConcreto

DecProcAbstrato ::= "abstract" AssinaturaMetodo

ListaDeclaracaoParametro ::= Tipo Id | Tipo Id "," ListaDeclaracaoParametro

Tipo ::= TipoClasse 

| TipoPrimitivo

TipoClasse ::= Id

TipoPrimitivo ::= "string" 

| "int" 

| "boolean"

modAcesso ::= "public" 

| "private"

